Laboratorní cvičení - Integrální počet v R

POZOR! Maple neuvádí ve výsledcích neurčitých integrálů integrační konstantu. Maple počítá integrály v oboru komplexních čísel.

Neurčitý integrál

Neurčitý integrál zapíšeme pomocí příkazu **Int**$(\mathbf{f}(\mathbf{x}), \mathbf{x})$, kde **f(x)**je integrovaná funkce a **x** je proměnná podle které integrujeme. Neurčitý integrál vypočteme pomocí příkazu **int**$(\mathbf{f}(\mathbf{x}), \mathbf{x})$, nebo použijeme příkaz $\int f \, \mathrm{d}x$ z lišty **Expression**.Neurčitý integrál zadaný příkazem **Int**$(\mathbf{f}(\mathbf{x}), \mathbf{x})$ je možné následovně vypočítat příkazem **value.**

Příklad 1. Vypočtěte integrál $\int (8 \cos x - 4 \sin x) \, \mathrm{d}x$.

```
> Int(8*cos(x)-4*sin(x),x)=int(8*cos(x)-4*sin(x),x);
```

$$\int (8 \cos(x) - 4 \sin(x)) \, \mathrm{d}x = 8 \sin(x) + 4 \cos(x)$$

Příklad 2. Vypočtěte integrál $\int x \cos x \, \mathrm{d}x$.

Tento integrál se řeší metodou per partes, Maple integrál spočítá, aniž bychom museli volit funkce $u(x)$ a $v'(x)$.

```
> Int(x*cos(x),x)=int(x*cos(x),x);
```

$$\int x \cos(x) \, \mathrm{d}x = \cos(x) + x \sin(x)$$

Příklad 3. Vypočtěte integrál $\int \frac{\mathrm{d}x}{\sqrt{1 - x^2} \arcsin x}$.

```
> Int(1/(sqrt(1-x^2)*arcsin(x)),x)=
int(1/(sqrt(1-x^2)*arcsin(x)),x);
```

$$\int \frac{1}{\sqrt{1 - x^2} \arcsin(x)} \, \mathrm{d}x = \ln(\arcsin(x))$$

Daný integrál by se řešil substitucí $t = \arcsin x.$ Všimněte si, že předchozí výsledek je správný pouze na otevřeném intervalu $(0, 1)$, na intervalu $(-1, 0)$ je správný výsledek $\ln(|\arcsin(x)|)$.

Příklad 4. Vypočtěte integrál $\int x e^{-x^2} \, \mathrm{d}x$.

```
> Int(x*exp(-x^2),x)=int(x*exp(-x^2),x);
```

$$\int x \, e^{-x^2} \, \mathrm{d}x = -\frac{1}{2} e^{-x^2}$$

Daný integrál by se řešil substitucí $t = -x^2$.

Maple vyřeší i (pro nás) složitější integrály, ve kterých je potřeba nejprve zavést substituci, a poté řešit metodou per partes.

Příklad 5. Vypočtěte integrál $\int x \sin(x^2 + 1) \, e^{-x^2} \, \mathrm{d}x$.

```
> Int(x*sin(x^2+1)*exp(-x^2),x)=int(x*sin(x^2+1)*exp(-x^2),x);
```

$$\int x \sin(x^2+1)\, e^{-x^2}\, dx$$
$$= \frac{1}{1+\tan\left(\frac{1}{2}x^2+\frac{1}{2}\right)^2}\left(-\frac{1}{2}e^{-x^2}\tan\left(\frac{1}{2}x^2+\frac{1}{2}\right) + \frac{1}{4}e^{-x^2}\tan\left(\frac{1}{2}x^2+\frac{1}{2}\right)^2 - \frac{1}{4}e^{-x^2}\right)$$

```
> simplify(rhs(%));
```

$$-\frac{1}{4}e^{-x^2}\left(2\sin\left(\frac{1}{2}x^2+\frac{1}{2}\right)\cos\left(\frac{1}{2}x^2+\frac{1}{2}\right) - 1 + 2\cos\left(\frac{1}{2}x^2+\frac{1}{2}\right)^2\right)$$

V předchozím příkaz **rhs(rovnice)** vrátí pravou stranu rovnice. Obdobný je příkaz **lhs(rovnice)**.

Maple lze použít i pro řešení integrálů z racionálně lomené funkce a dalších funkcí, které se řeší pomocí vhodných substitucí. Např.

Příklad 6. Vypočtěte integrál $\int \frac{2x+2}{(x-1)(x^2+1)^2}\, dx$.

```
> Int((2*x+2)/((x-1)*(x^2+1)^2),x)=
int((2*x+2)/((x-1)*(x^2+1)^2),x);
```

$$\int \frac{2x+2}{(x-1)(x^2+1)^2}\, dx = \frac{1}{x^2+1} - \frac{1}{2}\ln(x^2+1) - \arctan(x) + \ln(x-1)$$

Příklad 7. Vypočtěte integrály $\int \frac{dx}{2-\cos x}$, $\int \frac{1}{\sqrt{x}+x^{1/4}}\, dx$, $\int \frac{x \arcsin(x)}{\sqrt{1-x^2}}\, dx$.

```
> Int(1/(2-cos(x)),x)=int(1/(2-cos(x)),x);
```

$$\int \frac{1}{2-\cos(x)}\, dx = \frac{2}{3}\sqrt{3}\arctan\left(\tan\left(\frac{1}{2}x\right)\sqrt{3}\right)$$

Opět pozor na definiční obory. Zatímco integrand je funkce definovaná a spojitá na celé reálné ose, výsledek je správný jen na intervalu $(-\pi, \pi)$ nebo jiném intervalu, který z něj dostaneme posunutím o celočíselný násobek 2π. Globální vzorec dostaneme tzv. slepováním - viz skripta.

```
> Int(1/(sqrt(x)+root[4](x)),x)=int(1/(sqrt(x)+root[4](x)),x);
```

$$\int \frac{1}{\sqrt{x}+x^{1/4}}\, dx = -2\ln(x^{1/4}-1) + 2\ln(x^{1/4}+1) + \ln(x-1) - 4x^{1/4} + 2\sqrt{x} + \ln(\sqrt{x}-1) - \ln(\sqrt{x}+1)$$

```
> Int(x*arcsin(x)/sqrt(1-x^2),x)=int(x*arcsin(x)/sqrt(1-x^2),x);
```

$$\int \frac{x \arcsin(x)}{\sqrt{1-x^2}}\, dx = x - \arcsin(x)\sqrt{1-x^2}$$

Některé integrály však Maple není schopen vyřešit, přestože bychom byli schopni odhadnout, jakou substituci použít a příklad tak vyřešit. V takovém případě si pomůžeme načtením balíčku

Student[Calculus1],který má v sobě zabudované funkce **Hint**(nápověda, jakou úpravu v integrandu udělat, aby se integrand zjednodušil) a **Rule** (aplikace předchozí nápovědy).

Příklad 8. Vypočtěte integrál $\displaystyle\int \frac{2\,x + \frac{1}{2\sqrt{x}}}{\left(x^2+\sqrt{x}\right)^2+1}\,\mathrm{d}x.$

```
> Int((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x)=
int((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x);
```

$$\int \frac{2\,x + \frac{1}{2\sqrt{x}}}{\left(x^2+\sqrt{x}\right)^2+1}\,\mathrm{d}x =$$

$$\sum_{\_R = RootOf\left(16963840\,\_Z^8 + 62208\,\_Z^6 + 131072\,\_Z^5 + 4320\,\_Z^4 + 112\,\_Z^2 +\right.} \_R\ln\left(\sqrt{x} - \frac{3132323754840}{6436067}\,\_R^7 - \frac{2647715815875}{51488536}\,\_R^6 - \frac{51502767138}{6436067}\,\_R^5 - \frac{421763268249}{205954144}\,\_R^4 - \frac{8166018297}{12872134}\,\_R^3 - \frac{24612875433}{823816576}\,\_R^2 + \frac{526417421}{51488536}\,\_R - \frac{830509011}{3295266304}\right) + 4\Bigg($$

$$\sum_{\_R = RootOf\left(16963840\,\_Z^8 + 1064128\,\_Z^6 + 7168\,\_Z^5 + 25008\,\_Z^4 + 336\,\_Z^3 +\right.} \_R\ln\left(\sqrt{x} + \frac{10435839142339812416 0}{18014137003713}\,\_R^7 + \frac{13226700886242764 80}{18014137003713}\,\_R^6 + \frac{4030353427414035232}{18014137003713}\,\_R^5 + \frac{883839504947025 28}{18014137003713}\,\_R^4 + \frac{68273585434575320}{18014137003713}\,\_R^3 + \frac{1587147495192748}{18014137003713}\,\_R^2 + \frac{160734078200506}{6004712334571}\,\_R + \frac{5058826964423}{18014137003713}\right)\Bigg)$$

Řešení jsme sice obdrželi, ale výsledek rozhodně nevypadá tak, jak bychom očekávali. Využijeme balíčku **Student[Calculus1].**

```
> with(Student[Calculus1]):
> Int((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x);
```

$$\int \frac{2\,x + \frac{1}{2\sqrt{x}}}{\left(x^2+\sqrt{x}\right)^2+1}\,\mathrm{d}x$$

```
> Hint(%);
```

$$\left[change, x = u^2, u\right]$$

Připomeňme, že znak % slouží v Maplu jako rychlý odkaz na (časově) poslední výsledek, %% je odkaz na předposlední výsledek, %%% je odkaz na předpředposlední výsledek atd.
V nápovědě jsme obrželi, že máme použít substituci $x = u^2$.

```
> Rule[%](%%);
```

$$\int \frac{2x + \frac{1}{2\sqrt{x}}}{\left(x^2 + \sqrt{x}\right)^2 + 1}\, dx = \int \frac{4u^3 + 1}{u^8 + 2u^5 + u^2 + 1}\, du$$

Po provedení substituce se nám integrand "zjednoduší". Dále můžeme postupovat obdobně. Vždy budeme střídavě aplikovat příkazy **Hint** a **Rule**, až integrál vypočteme.
POZN.Maple si v substituci sám volí označení nových proměnných.

```
> Hint(%);
```

$$\left[change, u1 = u + u^4, u1\right]$$

```
> Rule[%](%%);
```

$$\int \frac{2x + \frac{1}{2\sqrt{x}}}{\left(x^2 + \sqrt{x}\right)^2 + 1}\, dx = \int \frac{1}{1 + u1^2}\, du1$$

```
> Hint(%);
```

$$\left[change, u1 = \tan(u2), u2\right]$$

```
> Rule[%](%%);
```

$$\int \frac{2x + \frac{1}{2\sqrt{x}}}{\left(x^2 + \sqrt{x}\right)^2 + 1}\, dx = \int 1\, du2$$

```
> Hint(%);
```

$$\left[constant\right]$$

```
> Rule[%](%%);
```

$$\int \frac{2x + \frac{1}{2\sqrt{x}}}{\left(x^2 + \sqrt{x}\right)^2 + 1}\, dx = u2$$

Výsledek integace dostaneme zpětným dosazením za jednotlivé proměnné. V našem případě je

$$\int \frac{2x + \frac{1}{2\sqrt{x}}}{\left(x^2 + \sqrt{x}\right)^2 + 1}\, dx = \arctan\left(x^2 + \sqrt{x}\right) + c.$$

Řešit daný integrál můžeme také pomocí zabudované funkce **IntTutor(f(x), x)**, která udělá předchozí výpočet, aniž bychom museli používat příkazů Hint a Rule, a navíc se po substitucích vrátí k původní proměnné. Při jejím použití ale ztrácíme možnost vidět jaké substituce (resp. úpravy) byly provedeny. Před použitím této funkce je potřeba vyvolat balíček **Student[Calculus1].**

```
> IntTutor((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x);
```

$$\int \frac{2x+\frac{1}{2\sqrt{x}}}{\left(x^2+\sqrt{x}\right)^2+1}\,dx = \arctan\left(x^2+\sqrt{x}\right)$$

Určitý integrál

Určitý integrál zapíšeme pomocí příkazu **Int**$(f(x), x = a..b)$, kde **f(x)**je integrovaná funkce a **a**, **b** jsou integrační meze. Určitý integrál vypočteme pomocí příkazu **int**$(f(x), x = a..b)$, nebo použijeme příkaz $\int_a^b f\,dx$ z lišty **Expression**.

Příklad 1. Vypočtěte integrál $\int_0^1 \frac{dx}{\sqrt{x^2+1}-x}$.

```
> Int(1/(sqrt(x^2+1)-x),x=0..1)=int(1/(sqrt(x^2+1)-x),x=0..1);
```

$$\int_0^1 \frac{1}{\sqrt{x^2+1}-x}\,dx = \frac{1}{2}+\frac{1}{2}\sqrt{2}-\frac{1}{2}\ln\left(\sqrt{2}-1\right)$$

Příklad 2. Vypočtěte integrál $\int_0^{\frac{\pi}{2}} \frac{10}{2\cos x+3}\,dx.$

```
> Int(10/(2*cos(x)+3),x=0..Pi/2)=int(10/(2*cos(x)+3), x = 0 ..
(1/2)*Pi);
```

$$\int_0^{\frac{1}{2}\pi} \frac{10}{2\cos(x)+3}\,dx = 4\arctan\left(\frac{1}{5}\sqrt{5}\right)\sqrt{5}$$

Jestliže Maple neumí daný integrál vypočítat, opíše zadání. V tomto případě je potřeba příklad vyřešit numericky pomocí příkazu **evalf(Int(f(x), x=a..b))**.
Maple použije vestavěnou numerickou metodu, kterou si sám zvolí (velmi přesnou).

Příklad 3. Vypočtěte integrál $\int_0^1 e^{\frac{x^2}{x^2+1}}\,dx$.

```
> Int(exp(x^2/(x^2+1)),x=0..1)=int(exp(x^2/(x^2+1)),x=0..1);
```

$$\int_0^1 e^{\frac{x^2}{x^2+1}}\,dx = \int_0^1 e^{\frac{x^2}{x^2+1}}\,dx$$

```
>
Int(exp(x^2/(x^2+1)),x=0..1)=evalf(int(exp(x^2/(x^2+1)),x=0..1));
```

$$\int_0^1 e^{\frac{x^2}{x^2+1}}\,dx = 1.255621168$$

Při výpočtech se můžeme setkat ještě s dalšími problémy, např. pokud nezadáme podmínku pro konstantu, kterou v integrandu použijeme, můžeme obdržet následující výsledek.

Příklad 4. Vypočtěte integrál $\int_0^1 x^n \, dx$.

Nejprve si uvědomme, že výsledek bude záviset na hodnotě *n*.Abychom si o něm udělali správnou představu, najdeme nejprve primitivní funkci k funkci $f(x) = x^n$.

```
> Int(x^n,x)=int(x^n,x);
```

$$\int x^n \, dx = \frac{x^{n+1}}{n+1}$$

```
> Int(x^n,x=0..1)=int(x^n,x=0..1);
```

$$\int_0^1 x^n \, dx = \frac{-\left(\lim_{x \to 0+} x^{n+1}\right) + 1}{n+1}$$

Je potřeba si uvědomit, že výsledek, který jsme obdrželi platí pouze pro $n \neq -1$. Výsledek můžeme zjednodušit, pokud použijeme podmínku pro *n*.

```
> assume(n>-1);
> Int(x^n,x=0..1)=int(x^n,x=0..1);
```

$$\int_0^1 x^{n\sim} \, dx = \frac{1}{n\sim + 1}$$

Označení $n\sim$ znamená, že o n byl učiněn nějaký předpoklad.

```
> assume(n<-1);
> Int(x^n,x=0..1)=int(x^n,x=0..1);
```

$$\int_0^1 x^{n\sim} \, dx = \infty$$

Jde o nevlastní divergentní integrál.

Nevlastní integrál

Nevlastní integrál se v Maple počítá stejně jako určitý. V případě, že Maple opisuje zadání, je opět potřeba použít numerického řešení pomocí příkazu **evalf.**

Integrál na neohraničeném intervalu

Příklad 1. Vypočtěte integrál $\int_1^\infty \frac{2}{x^3} \, dx$.

```
> Int(2/x^3,x=1..infinity)=int(2/x^3,x=1..infinity);
```

$$\int_1^\infty \frac{2}{x^3} \, dx = 1$$

Daný integrál lze vypočítat z definice nevlastního integrálu, a to limitou.

```
> Int(2/x^3,x=1..infinity)=
Limit(int(2/x^3,x=1..t),t=+infinity);
Warning, unable to determine if 0 is between 1 and t; try to use
assumptions or use the AllSolutions option
```

$$\int_1^\infty \frac{2}{x^3} \, dx = \lim_{t \to \infty} \left(\int_1^t \frac{2}{x^3} \, dx \right)$$

Předchozí varování znamená, že Maple není schopen spočítat hodnotu určitého integrálu s horní mezí t bez informace o t. Volba **AllSolutions** způsobí, že se vypíše výsledek v závislosti na t.

```
> Int(2/x^3,x=1..infinity)=
Limit(int(2/x^3,x=1..t, AllSolutions),t=+infinity);
```

$$\int_1^{\infty} \frac{2}{x^3}\, dx = \lim_{t \to \infty} \left(\begin{cases} \mathit{undefined} & t < 0 \\ -\infty & t = 0 \\ \dfrac{-1 + t^2}{t^2} & 0 < t \end{cases} \right)$$

Pro vyčíslení limity na pravé straně použijeme příkaz **value** (**rhs**(**%**)) , **value** provede vyhodnocení celého předchozího zápisu, příkaz **rhs**(right hand side) pouze předá pravou stranu předchozí rovnice.

```
> value(rhs(%));
```

$$1$$

Integrovanou funkci si pro lepší představu můžeme vykreslit.

```
> plot(2/x^3,x=1..infinity);
```



Pozor! Měřítko na ose x předchozího obrázku není lineární, koncový bod odpovídá nekonečnu. Obrázek je tudíž poněkud zavádějící.

Integrál z neohraničené funkce

Příklad 1. Vypočtěte integrál $\displaystyle\int_0^2 \frac{1}{2-x}\, dx$.

Funkce je zřejmě neohraničená, a to v bodě $x=2$, pro lepší představu si nakreslíme její graf.

```
> plot(1/(2-x),x=0..2);
```




```
> Int(1/(2-x),x=0..2)=int(1/(2-x),x=0..2);
```

$$\int_0^2 \frac{1}{2-x}\,dx = \infty$$

Nyní výpočet opět provedeme z definice pomocí limity.

```
> Int(1/(2-x),x=0..2)=Limit(int(1/(2-x),x=0..t),t=2,left);
Warning, unable to determine if 2 is between 0 and t; try to use assumptions or use the AllSolutions option
```

$$\int_0^2 \frac{1}{2-x}\,dx = \lim_{t\to 2^-} \left(\int_0^t \frac{1}{2-x}\,dx\right)$$

Předchozí varování znamená, že Maple není schopen spočítat hodnotu určitého integrálu s horní mezí t bez informace o t. Volba **AllSolutions** způsobí, že se vypíše výsledek v závislosti na t.

```
> Int(1/(2-x),x=0..2)=Limit(int(1/(2-x),x=0..t,
AllSolutions),t=2,left);
```

$$\int_0^2 \frac{1}{2-x}\,dx = \lim_{t\to 2^-} \left(\begin{cases} \ln(2)-\ln(2-t) & t\le 0 \\ \ln(2)+\mathrm{I}\,\pi-\ln(t-2) & t<2 \\ \infty & t=2 \\ \mathit{undefined} & 2<t \end{cases}\right)$$

```
> value(rhs(%));
```

$$\infty$$

Aplikace

Délka křivky

Obecný vzorec pro výpočet délky křivky je **ArcLength(f(x),x=a..b),** pokud je křivka zadaná parametricky, provede se výpočet pomocí příkazu **ArcLength([f(x), g(x)], x=a..b).**

```
> with(Student[Calculus1]):
> ArcLength(f(x),x=a..b);
```

$$\int_a^b \sqrt{\left(\frac{d}{dx} f(x)\right)^2 + 1}\, dx$$

```
> ArcLength([f(x), g(x)], x=a..b);
```

$$\int_a^b \sqrt{\left(\frac{d}{dx} f(x)\right)^2 + \left(\frac{d}{dx} g(x)\right)^2}\, dx$$

Příklad 1. Vypočtěte délku oblouku paraboly $y = x^2, 0 \le x \le 3$.

```
> ArcLength(x^2, x=0..3,output=integral);
```

$$\int_0^3 \sqrt{4x^2 + 1}\, dx$$

```
> ArcLength(x^2, x=0..3);
```

$$\frac{3}{2}\sqrt{37} - \frac{1}{4}\ln\left(-6 + \sqrt{37}\right)$$

```
> ArcLength(x^2, x=0..3,output=plot);
```



The arc length of $f(x) = x^2$ on the interval $[0, 3]$. The coordinate system is cartesian

Příklad 2. Určete délku oblouku rovinné křivky $x = a(t - \sin t), y = a(1 - \cos t), a > 0.$

```
> assume(a>0);
> ArcLength([a*(t-sin(t)),a*(1-cos(t))],
t=0..2*Pi,output=integral);
```

$$\int_0^{2\pi} a\sim \sqrt{1 - 2\cos(t) + \cos(t)^2 + \sin(t)^2}\,\mathrm{d}t$$

```
> ArcLength([a*(t-sin(t)),a*(1-cos(t))], t=0..2*Pi);
```

$$8\,a\sim$$

Předchozí křivka je tzv. *cykloida*. Nakreslíme si ji např. pro $a = 2$.

```
> a:=2;
```

$$a := 2$$

```
> plot([a*(t-sin(t)),a*(1-cos(t)), t=0..2*Pi]);
```



Objem rotačního tělesa

Obecný vzorec pro výpočet objemu rotačního tělesa je v případě, že křivočarý obdélník, jehož rotací vznikne těleso, ohraničen jedinou funkcí f(x),
VolumeOfRevolution(f(x), x=a..b), v případě dvou funkcí je obecný vzorec
VolumeOfRevolution(f(x),g(x), x=a..b).

```
> with(Student[Calculus1]):
> VolumeOfRevolution(f(x), x=a..b);
```

$$\int_{2}^{b} \pi f(x)^2 \, dx$$

```
> VolumeOfRevolution(f(x),g(x), x=a..b);
```

$$\int_{2}^{b} \pi \left| f(x)^2 - g(x)^2 \right| dx$$

Příklad 1. Vypočtěte objem tělesa, které vznikne rotací obrazce omezeného křivkami
$y = \frac{1}{1+x^2}, x = -1, x = 1.$

```
> VolumeOfRevolution(1/(1+x^2),x=-1..1,output=integral);
```

$$\int_{-1}^{1} \frac{\pi}{\left(x^2+1\right)^2} \, dx$$

```
> VolumeOfRevolution(1/(1+x^2),x=-1..1);
```

$$\frac{1}{4}\pi^2 + \frac{1}{2}\pi$$

```
> VolumeOfRevolution(1/(1+x^2),x=-1..1,output=plot);
```

The Volume of Revolution Around the Horizontal Axis of
f(x) = 1/(x^2+1)
on the Interval [-1, 1]



```
> VolumeOfRevolution(1/(1+x^2),x=-1..1,output=animation);
```



Příklad 2. Vypočtěte objem tělesa, které vznikne rotací obrazce omezeného křivkami $y^2 = x, y = x^2, y \geq 0$.

```
> solve(sqrt(x)=x^2);
```

$$0, 1$$

```
> VolumeOfRevolution(sqrt(x),x^2,x=0..1,output=integral);
```

$$\int_0^1 \pi \left| -x + x^4 \right| \mathrm{d}x$$

```
> VolumeOfRevolution(sqrt(x),x^2,x=0..1);
```

$$\frac{3}{10}\pi$$

```
> VolumeOfRevolution(sqrt(x),x^2,x=0..1,output=animation);
```



Obsah pláště rotačního tělesa

Obsah pláště tělesa vypočteme pomocí příkazu **SurfaceOfRevolution(f(x), x=a..b).**

```
> with(Student[Calculus1]):
> SurfaceOfRevolution(f(x), x=a..b);
```

$$\int_2^b 2\pi\,|f(x)|\sqrt{\left(\frac{d}{dx}f(x)\right)^2+1}\,dx$$

Příklad. Vypočtěte obsah pláště tělesa, které vznikne rotací křivky $f(x) = 4 + x, x \in \langle -4, 2\rangle$.

```
> SurfaceOfRevolution(4+x,x=-4..2);
```

$$36\,\pi\sqrt{2}$$

```
> SurfaceOfRevolution(4+x,x=-4..2,output=integral);
```

$$\int_{-4}^{2} 2\pi\,(4+x)\sqrt{2}\,dx$$

```
> SurfaceOfRevolution(4+x,x=-4..2,output=plot);
```

The Surface of Revolution Around the Horizontal Axis of
f(x) = 4+x
on the Interval [-4, 2]



Úlohu lze řešit i interaktivně:

```
> SurfaceOfRevolutionTutor(4+x,x=-4..2);
```




```
>
```

Numerické metody

Pro numerické výpočty integrálů použijeme zabudovaných funkcí. Nejprve načteme balíček **Student[Calculus1].**V tomto balíčku je možné vyvolat příkazem **RiemannSum(f(x),x=a..b,opt)** jednu z metod pro přibližný výpočet integrálu (obdélníková metoda).Tato metoda rozdělí daný interval na 10 podintervalů (pokud nezvolíme jiné dělení) a v každém podintervalu vypočte funkční hodnotu ve vybraném bodě (viz nápověda), poté vypočte obsah obrazce ohraničeného grafem funkce podle vzorce $S=\sum_{i=1}^{N} f\left(x_i^*\right)\left(x_i-x_{i-1}\right)$. Volbou parametru **opt** můžeme výstup získat jako animaci nebo obrázek.

Příklad 1. Vypočtěte $\int_{-2}^{2} x^2 \cos x \, dx$.

```
> with(Student[Calculus1]):
> RiemannSum(x^2*cos(x), x=-2..2, output = animation);
```



Poklepáním na obrázek se dostaneme do menu pro spuštění animace. Je vhodné animaci zpomalit, např. navolit FPS=1.

```
> RiemannSum(x^2*cos(x), x=-2..2, output = plot);
```



Dále můžeme pro přibližný výpočet integrálu použít příkaz **ApproximateInt(f(x),x=a..b,opt).**

```
> ApproximateInt(x^2*cos(x),
x=-2..2,method=trapezoid,output=plot,partition=10);
```




```
> ApproximateInt(x^2*cos(x),
x=-2..2,method=simpson,output=plot,partition=4);
```



Všimněte si, že když v předchozím příkazu zvýšíte parametr **partition**na 6 nebo více, téměř nevidíte pouhým okem rozdíl mezi zadanou funkcí a její aproximací kubickými křivkami.

```
> S:=Int(x^2*cos(x), x=-2..2);
```

$$S := \int_{-2}^{2} x^2 \cos(x)\, dx$$

```
> ApproximateInt(S);
```

$$\frac{324}{125}\cos\left(\frac{9}{5}\right) + \frac{196}{125}\cos\left(\frac{7}{5}\right) + \frac{4}{5}\cos(1) + \frac{36}{125}\cos\left(\frac{3}{5}\right) + \frac{4}{125}\cos\left(\frac{1}{5}\right)$$

```
> evalf(%);
```

$$0.3789012829$$

```
> int(x^2*cos(x), x=-2..2);
```

$$4\sin(2) + 8\cos(2)$$

```
> evalf(%);
```

$$0.308015015$$

Všimněte si rozdílných přibližných hodnot daného integrálu.

Pokud nastavíme parametr opt na plot, dostaneme náhradu integrálu pomocí obdélníkové metody.

Další podrobnosti viz nápověda **Student[Calculus1][ApproximateInt].**

```
> ApproximateInt(S,output=plot);
```



Pro numerický výpočet integrálů můžeme také použít interaktivní tutor, který otevřeme z horní lišty **Tools-Tutor-Calculus-single variable-Approximate Integration.**Po otevření dostaneme okno, do kterého si můžeme zapsat danou funkci, interval a vybrat si jednu z nabízených numerických metod, pro nás připadá v úvahu **Riemann rule**(jako obdélníková metoda, lze ale volit z více možností, jak vybrat reprezentanty dílků dělení), **Trapeziodal rule**(lichoběžníková metoda) nebo **Simpson's rule** (Simpsonova metoda).

```
> Student[Calculus1][ApproximateIntTutor](x^2*cos(x), x=-2..2);
```



V tomto případě byla vybrána lichoběžníková metoda.

```
> Student[Calculus1][ApproximateIntTutor](x^2*cos(x), x=-2..2);
```



V tomto případě bylo vybráno Simpsonovo pravidlo.